



Vieweg+TeubnerPLUS

Zusatzinformationen zu Medien des Vieweg+Teubner Verlags

Ausgleichsvorgänge in elektro-mechanischen Systemen mit Maple analysieren

Grundwissen für Antriebstechnik und Mechatronik

2011 | 1. Auflage

Kreisdiagramm eines Asynchronmotors mit Maple konstruieren

Ein Kreisdiagramm, d. h. die Ortskurve des Ständerstromes, ermöglicht es, das Verhalten eines Drehstrom-Asynchronmotors bei veränderlicher Belastung schnell zu überblicken. Für einen Asynchronmotor mit Schleifringläufer soll es mit Hilfe von Maple konstruiert werden. Verwendet wird die vereinfachte Form (Heyland-Diagramm). Bezüglich der theoretischen Grundlagen der Konstruktion des Kreisdiagramms wird auf [2] und [3] verwiesen. Die folgenden Daten sind gegeben [1]:

$U_n = 500 \text{ V}$, $f = 50 \text{ Hz}$, $I_{1,n} = 28,7 \text{ A}$, $N = 18 \text{ kW}$, $n = 967,5 \text{ min}^{-1}$, $\cos\phi = 0,844$

Rotor: $U_{2,n} = 83,2 \text{ V}$, Strangwiderstände der Wicklungen: $R_1 = 0,465 \Omega$, $R_2 = 0,00985 \Omega$,

Stern-/Sternschaltung

Leerlauf: $I_{1,0} = 8,5 \text{ A}$, $\cos\phi_0 = 0,15$;

Kurzschlussversuch: $U_k = 170,5 \text{ V}$ bei $I_{1,k} = 28,7 \text{ A}$, $\cos\phi_k = 0,277$

Lösungsweg

Die Parameter des Kreises werden aus drei Kreispunkten P_0 , P_n und P_k , die durch die komplexen Statorströme (Betrag und Winkel) für die Zustände Leerlauf, Nennbetrieb und Kurzschluss definiert werden, berechnet. Diese Punkte der gesuchten Ortskurve liefern unter Verwendung der allgemeinen Kreisgleichung ein System von drei quadratischen Gleichungen zur Bestimmung der Koordinaten des Mittelpunktes und des Radius des Kreises. Der Heylandkreis wird im Maple-Worksheet dargestellt, dann durch die Eintragung des Punktes P_{kipp} ergänzt und nach einer Auswertung auch noch ausgedruckt. Durch die Einführung der zusätzlichen Variablen P_0 , P_n und P_k zur Bezeichnung der charakteristischen Punkte wird lediglich die Schreibweise der Gleichungen vereinfacht.

Ein zweiter Lösungsweg, für den lediglich der Leerlaufpunkt und der Kurzschlusspunkt bekannt sein müssen, wird im Maple-Worksheet "Kreisdiagramm Beispiel 2" verwendet.

Literatur

- [1] Bläss, H.: Definitionsbeispiel zur Kreisdiagrammkonstruktion
- [2] Bödefeld/Sequenz: Elektrische Maschinen. 7. Aufl., Springer Verlag 1965, S. 213 ff.
- [3] Grafe, K.: Asynchronmotoren. Fachbuchverlag Leipzig 1953. S. 54 ff.
- [4] Müller, R.: Ausgleichsvorgänge in elektro-mechanischen Systemen mit Maple analysieren. Vieweg+Teubner Verlag 2011

Maple-Programm

```
> restart: with(plots):  
> interface(imaginaryunit=j, displayprecision=4):  
> plotsetup("inline", plotoutput=terminal, plotoptions="colour=cmk,  
resolution=2000"):
```

Gegebene Motordaten:

Nennbetrieb:

```
> U[n] := 500:      # V
> U[2,n] := 83.2:   # V
> I[1,n] := 28.7:   # A
> phi[n] := arccos(0.844):
> evalf(180/Pi*phi[n]); # Grad
32.4350

> n[0] := 1000:     # 1/min
> R[1] := 0.465:   R[2] := 0.00985:   # Ohm
> ü := U[n]/U[2,n]; # Übersetzungsverhältnis
ü := 6.0096
```

Leerlauf:

```
> I[1,0] := 8.5:   # A
> phi[0] := arccos(0.15):
> evalf(180/Pi*phi[0]); # phi[0] in Grad
81.3731
```

Kurzschlussversuch:

```
> U[k] := 170.5:   # V
> I[k] := 28.7:   # A
> phi[k] := arccos(0.277):
> evalf(180/Pi*phi[k]); # phi[k] in Grad
73.9188
```

1 Berechnung der Punkte P_n , P_0 und P_k des Kreisdiagramms

Im Folgenden werden indizierte Maple-Variablen, die Zeigergrößen repräsentieren, durch einen Unterstrich zwischen Variablennamen und Index gekennzeichnet. Da negative Indizes ausgeschlossen sind, kann die gewählte Darstellung nicht zu Missverständnissen führen. Einen Unterstrich hinter der Index-Angabe lässt Maple nicht zu und ein Unterstrich als erstes Zeichen eines Namens soll Variablen der Maple-Bibliothek vorbehalten bleiben.

Einige Teile von Befehlen werden mit Apostroph-Zeichen eingeschlossen, um deren sofortige Auswertung unter Verwendung vorher zugewiesener Variablenwerte zu unterbinden.

Nennpunkt P_n Zeiger $I_{1,n}$:

```
> I_1,n := 'I[1,n]*exp(j*phi[n])';
```

$$I_{\underline{1},n} := I_{1,n} e^{j\phi_n}$$

Zeiger auf den Punkt P_n :

```
> Pn_:= I_[1,n] :
```

Leerlaufpunkt P_0

Zeiger $I_{1,0}$, Zeiger auf den Punkt P_0 :

```
> I_[1,0]:= 'I[1,0]*exp(j*phi[0])' ; P0_:= I_[1,0] :
```

$$I_{-1,0} := I_{1,0} e^{j\phi_0}$$

Kurzschlusspunkt P_k

Mit der beim Kurzschlussversuch ermittelten Kurzschlussspannung U_k wird der Effektivwert des Kurzschlussstromes bei Nennspannung bestimmt.

```
> I[1,k]:= 'I[1,n]*U[n]/U[k]' ;
```

$$I_{1,k} := \frac{I_{1,n} U_n}{U_k}$$

Zeiger des Kurzschlussstroms $I_{1,k}$ und Zeiger auf den Punkt P_k :

```
> I_[1,k]:= 'I[1,k]*exp(j*phi[k])' ; Pk_:= I_[1,k] :
```

$$I_{-1,k} := I_{1,k} e^{j\phi_k}$$

2 Bestimmung der Kreisparameter

Bestimmungsgleichungen für die Kreisparameter:

Es sind xm der Abszissenwert und ym der Ordinatenwert des Kreismittelpunktes sowie r der Radius des Kreises.

```
> G1:= '(Im(P0_)-xm)^2+(Re(P0_)-ym)^2 = r^2' ;
```

$$G1 := (\Im(P0_-) - xm)^2 + (\Re(P0_-) - ym)^2 = r^2$$

```
> G2:= '(Im(Pk_)-xm)^2+(Re(Pk_)-ym)^2 = r^2' ;
```

$$G2 := (\Im(Pk_-) - xm)^2 + (\Re(Pk_-) - ym)^2 = r^2$$

```
> G3:= '(Im(Pn_)-xm)^2+(Re(Pn_)-ym)^2 = r^2' ;
```

$$G3 := (\Im(Pn_-) - xm)^2 + (\Re(Pn_-) - ym)^2 = r^2$$

Ermittlung der Lösungen des Gleichungssystems:

```
> Loe:= solve({G1,G2,G3}, [xm,ym,r]) ;
```

```
Loe := [[xm = 47.8270, ym = 1.8061, r = 39.4267], [xm = 47.8270, ym = 1.8061, r = -39.4267]]
```

```
> assign(Loe) ;
```

3 Darstellung von Heylandkreis und charakteristischen Punkten

Definition von Prozeduren

Für die Darstellung der komplexen Zeiger als Linie wird die Prozedur *linie* definiert. Diese erzeugt eine Plot-Struktur für eine Linie zwischen Anfangs- und Endpunkt eines Zeigers. Dabei werden die Punkte als komplexe Größen vorgegeben, deren Imaginärteil den Abszissenabschnitt und deren Realteil den Ordinatenwert bezeichnet.

```
> linie:= proc(ap, ep, opt)
    # ap... Anfangspunkt, ep...Endpunkt
    # opt...Optionen (z.B. Farbe, Dicke); als Liste vorzugeben
    plot([ [Im(ap), Re(ap)], [Im(ep), Re(ep)] ], op(opt));
end proc;
```

Die Prozedur *punkt* dient der graphischen Darstellung eines Punktes am Endpunkt eines Zeigers.

```
> punkt:= proc(pp)
    # pp... Zeiger auf den Punkt
    pointplot([ [Im(pp), Re(pp)] ], symbol=solidcircle);
end proc;
```

Plot-Strukturen für den Heylandkreis und die Punkte P_0 , P_n und P_k

```
> kreis:= plottools[circle]([xm,ym], r, color=black):
> P0:= punkt(P0_): Pk:= punkt(Pk_): Pn:= punkt(Pn_):
```

Konstruktion von Drehmoment- und Leistungslinie

Das Lot vom Punkt P_k schneidet die Abszissenparallele durch P_0 im Punkt D . Für den Zeiger vom Koordinatenursprung auf den Punkt D folgt demnach

```
> D_:= 'Re(P0_)+j*Im(Pk_)';

$$D_ := \Re(P0_) + j \Im(Pk_)$$

```

Die Strecke $D-P_k$

Im Punkt P_k wird keine mechanische Leistung abgegeben. Die Strecke $D-P_k$ ist daher ein Maß für die Verlustleistung bei stillstehendem Rotor (Schlupf $s = 1$). Diese setzt sich im Wesentlichen aus den Stromwärmeverlusten in Stator und Rotors zusammen. Durch Aufteilung der Strecke $D-P_k$ gemäß den Anteilen von Stator und Rotor an der Gesamtheit der Stromwärmeverlustleistung im Kurzschlusspunkt ergibt sich der Punkt E . Die Aufteilung erfolgt unter den in [2] beschriebenen Annahmen nach dem Anteil des Widerstands R_1 und des auf die Primärseite bezogenen Rotorwiderstands R_2' am Gesamtwiderstand. Es ergeben sich so die Teilstrecke $D-E$, die den Verlustanteil des Stators bezeichnet, und die Teilstrecke $E-P_k$ für die Stromwärmeverlustleistung im Rotor.

Zeiger des auf Primärseite reduzierten Rotorstroms im Kurzschlusspunkt:

```
> I_[2,k] := 'I_[1,k]-I_[1,0]';
```

$$I_{-2,k} := I_{-1,k} - I_{-1,0}$$

Strecke $D-E$:

```
> D_E := 'Re(I_[2,k])*R[1]/(R[1]+R[2]*ü^2)';
```

$$D_E := \frac{\Re(I_{-2,k}) R_1}{R_1 + R_2 \ddot{u}^2}$$

Zeiger vom Koordinatenursprung auf den Punkt E :

```
> E_ := 'D_ + D_E';
```

$$E_- := D_- + D_E$$

Zeiger vom Punkt P_0 auf den Punkt E :

```
> EE_ := 'E_ - P0_';
```

$$EE_- := E_- - P0_-$$

Drehmoment- und Leistungslinie

Die Gerade vom Leerlaufpunkt P_0 durch den Punkt E wird als Drehmomentlinie bezeichnet. Sie schneidet den Heylandkreis im Unendlichkeitspunkt P_∞ , d. h. an dem Punkt, dem der Schlupf mit dem Wert ∞ zugeordnet ist.

```
> ML:= linie(P0_, E_, [legend="Drehmomentlinie ", color=blue]):
```

Die Gerade von P_0 nach P_k bildet die Leistungslinie LL :

```
> LL:= linie(P0_, Pk_, [linestyle=dash, legend="Leistungslinie ",
color=blue]):
```

Unter Verwendung der oben definierten Zeiger werden nun noch Plotstrukturen für die Darstellung weiterer charakteristischer Punkte und Linien im Kreisdiagramm erzeugt.

```
> E:= punkt(E_):
> DD:= punkt(D_): # Name D ist für den Differentialoperator reserviert
> DPk:= linie(D_, Pk_, [color=black]):
> P0D:= linie(P0_, D_, [color=black]):
```

Darstellung des Kreisdiagramms

Plot-Struktur für Zeiger $I_{1,0}$:


```
> I_10:= linie(0, P0_, [color=blue]):
```

Konstante für Abstand der Beschriftungen im Kreisdiagramm:

```
> d:= 0.05*Im(Pk_): # Textabstand
> text:= textplot([[Im(P0_)-0.7*d, Re(P0_)+d/2,"P0"],
[Im(Pn_)-0.7*d, Re(Pn_)+d/2,"Pn"], [Im(Pk_)+d, Re(Pk_), "Pk"],
[Im(D_)+d/2, Re(D_)+1,"D"], [Im(E_)+d/2, Re(E_)+1,"E"]]):
```

```
> KD:= display(kreis, P0, Pk, Pn, DD, E, I_10, LL, ML, P0D, DPk, text,
  scaling=constrained, title=typeset("Asynchronmotor\n U =
  ",U[n], " V, I = ",I[1,n]," A\n"), titlefont=[TIMES,12,BOLD],
  labels=[typeset(Im('I[1]/A')),typeset(Re('I[1]/A'))]): KD;
```

Asynchronmotor
U = 500 V, I = 28.7000 A



4 Ermittlung des Kipppunktes

Den Kipppunkt, d. h. den Punkt des maximalen Drehmoments, erhält man, wenn man eine Parallele zur Drehmomentlinie als Tangente an den oberen Halbkreis legt.

Steigung der Drehmomentlinie = $\text{Re}(EE_-)/\text{Im}(EE_-)$; EE_- ... Zeiger von P_0 auf den Punkt E

Gleichung einer Tangente am Punkt (x_k, y_k) des Kreises:

```
> G4:= '(x-xm)*(xk-xm) + (y-ym)*(yk-ym) = r^2';
```

$$G4 := (x - xm)(xk - xm) + (y - ym)(yk - ym) = r^2$$

Auflösung der Gleichung $G4$ nach y ; Erzeugung der Form $y = Ax + B$:

```
> G5:= isolate(G4,y);
```

$$G5 := y = \frac{r^2 - (x - xm)(xk - xm)}{yk - ym} + ym$$

```
> G6:= collect(G5, x);
```

$$G6 := y = \frac{(-1.0000xk + 47.8270)x}{yk - 1.8061} + \frac{-732.9530 + 47.8270xk}{yk - 1.8061} + 1.8061$$

Terme des Ausdrucks zur Berechnung von y ermitteln:

```
> AA := op(rhs(G6));
```

$$AA := \frac{(-1.0000xk + 47.8270)x}{yk - 1.8061}, \frac{-732.9530 + 47.8270xk}{yk - 1.8061}, 1.8061$$

Der 1. Term von AA enthält den Koeffizienten von x . Dieser wird herausgelöst, mit der Steigung der Drehmomentlinie gleichgesetzt und nach yk aufgelöst.

```
> A := AA[1]/x;
```

$$A := \frac{-1.0000xk + 47.8270}{yk - 1.8061}$$

```
> G7 := isolate(A = Re(EE_)/Im(EE_), yk);
```

$$G7 := yk = -5.8038xk + 279.3827$$

Die Kreisgleichung lautet

```
> G8 := (xk-xm)^2 + (yk-ym)^2 = r^2;
```

$$G8 := (xk - 47.8270)^2 + (yk - 1.8061)^2 = 1554.4657$$

Das Gleichungssystem {G7, G8} hat zwei Lösungen:

```
> Loe2 := solve({G7,G8}, [xk,yk]);
```

$$Loe2 := [[xk = 41.1323, yk = 40.6603], [xk = 54.5216, yk = -37.0480]]$$

Nur die erste Lösung entspricht dem gesuchten Punkt P_{kipp} .

Zeiger auf P_{kipp} :

```
> I_[1,kipp] := subs(Loe2[1], j*xk + yk); Pkipp_ := I_[1,kipp];
```

$$I_{-1, kipp} := 40.6603 + 41.1323j$$

Plot-Struktur für Kippunkt P_{kipp} erzeugen:

```
> Pkipp := punkt(Pkipp_);
```

5 Rechnerische Auswertung des Kreisdiagramms

Bestimmung der Maßstäbe für die Leistung und das Drehmoment

```
> LMS := sqrt(3)*U[n]/1000; # Leistungsmaßstab in kW/A
```

$$LMS := \frac{1}{2} \sqrt{3}$$

```
> DMS := sqrt(3)*U[n]/(2*Pi*n[0]/60); # Drehmomentmaßstab in Nm/A
```

$$DMS := \frac{15\sqrt{3}}{\pi}$$

Prozedur zur Ermittlung des Drehmoments am Punkt P:

```
> moment := proc(P_)
```

```
    local S; global P0_, E_, D_, DMS;
```

```
    # Ermittlung der Strecke von P bis zur Drehmomentlinie
```



```

S:= Re(P_) - Re(P0_) - (Im(P_) - Im(P0_)) * (Re(E_) - Re(D_)) /
    (Im(D_) - Im(P0_));
evalf(S*DMS);
end proc:

```

Prozedur zur Ermittlung der mechanischen Leistung für Punkt P:

```

> leistung:= proc(P_)
    local S; global P0_, Pk_, D_, LMS;
    # Ermittlung der Strecke von P bis zur Leistungslinie
    S:= Re(P_) - Re(P0_) - (Im(P_) - Im(P0_)) * (Re(Pk_) - Re(D_)) /
        (Im(D_) - Im(P0_));
    evalf(S*LMS);
end proc:

```

Auswertung für den Nennpunkt Pn

Drehmoment:

```

> M[n] := moment(Pn_);

```

$$M_n := 179.8177$$

Mechanische Leistung in kW:

```

> Pmech[n] := leistung(Pn_);

```

$$Pmech_n := 18.0326$$

Auswertung für den Kippunkt Pkipp

Drehmoment in Nm:

```

> M[kipp] := moment(Pkipp_);

```

$$M_{kipp} := 279.0783$$

Mechanische Leistung in kW:

```

> Pmech[kipp] := leistung(Pkipp_);

```

$$Pmech_{kipp} := 25.4889$$

6 Drucken des vervollständigten Kreisdiagramms

Das zu druckende Diagramm soll im Postskriptformat in der Datei *print.ps* gespeichert werden. Mit dem Befehl **plotsetup** wird diese Form der Ausgabe vorbereitet. In Anpassung an die übliche Darstellung des Kreisdiagramms werden die Achsenbezeichnungen und auch die Achsenmarkierungen bei der Ausgabe unterdrückt.

```

> text2:= textplot([[Im(Pkipp_), Re(Pkipp_)+d, "Pkipp"]]):
> plotsetup(ps, plotoutput="print", plotoptions="noborder,
    resolution=2000"):

```

Ausgabe des Diagramms in die angegebene Datei *print.ps*:

```

> display(KD, Pkipp, text2, labels=[" ", " "], tickmarks=[0,0],
    caption=" ");

```

Nach dem Ausdrucken des in der Datei *print.ps* gespeicherten Kreisdiagramms sind für dessen weitere Auswertung der Strom-, der Drehmoment- und der Leistungsmaßstab erforderlich. Zu deren Berechnung wird die aus dem Diagramm ersichtliche Länge des Zeigers \underline{P}_k mit dem Befehl **readstat** angefordert.

Ermittlung des Strommaßstabs:

```
> ZL:= readstat("Zeigerlänge vom Punkt (0,0) bis Pk in mm:");
```

ZL := 142

```
> IMSd:= abs(Pk_)/ZL; # A/mm
```

IMSd := 0.5927

Berechnung des Leistungs- und des Drehmomentmaßstabs:

```
> LMSd:= evalf(IMSd*U[n]*sqrt(3)/1000); # kW/mm
```

LMSd := 0.5133

```
> DMSd:= evalf(LMSd*1000/(2*Pi*n[0]/60)); # Nm/mm
```

DMSd := 4.9016

Ausgleichsvorgänge in elektro-mechanischen Systemen
mit Maple analysieren

Grundwissen für Antriebstechnik und Mechatronik

Müller, R.

2011, XII, 284 S. 69 Abb. Mit zahlreichen Beispielen und

Maple-Plots., Softcover

ISBN: 978-3-8348-1217-9